Slide 1

AMPL Data File for an MCNFP Instance

```
set NODES := 1 2 3 4;
set ARCS  := (1,2) (1,3) (2,3) (2,4) (3,4);
param b :=
1  5
2 -2
3  0
4 -3;

#    cost    lower   upper bound
param:  c     l       u :=
1 2     3     2       5
1 3     2     0       2
2 3     1     0       2
2 4     4     1       3
3 4     4     0       3;
```

Slide 2

AMPL Model File for an MCNFP Instance

```
# AMPL model for the Minimum Cost Network Flow Problem
#
# By default, this model assumes that b[i] = 0, c[i,j] = 0,
# l[i,j] = 0 and u[i,j] = Infinity.
#
# Parameters not specified in the data file
# will get their default values.
set NODES;              # nodes in the network

set ARCS within {NODES, NODES};     # arcs in the network
```

This line says that $ARCS \subseteq NODES \times NODES$.

ARCS is a set of ordered pairs of elements of the set NODES.

For the example if $NODES = \{1, 2, 3, 4\}$, then
$ARCS \subseteq \{(1,1), (1,2), (1,3), (1,4), \ldots, (4,1), (4,2), (4,3), (4,4)\}$.

Slide 3

```
param b {NODES} default 0;
# b[i] = supply/demand for node i

param c {ARCS} default 0;
# c[i,j] = cost of one of flow on arc(i,j)

param l {ARCS} default 0;
# lower bound on flow on arc(i,j)

param u {ARCS} default Infinity;
# upper bound on flow on arc(i,j)

var x {ARCS};
# flow on arc (i,j)
```

Slide 4

```
minimize cost:
sum{(i,j) in ARCS} c[i,j] * x[i,j];
```

The objective function is equivalent to

$$\min \sum_{(i,j) \in A} c_{ij} x_{ij}.$$

```
# Flow Out(i) - Flow In(i) = b(i)
subject to flow_balance {i in NODES}:
sum{j in NODES: (i,j) in ARCS} x[i,j] -
sum{j in NODES: (j,i) in ARCS} x[j,i] = b[i];

subject to capacity {(i,j) in ARCS}:
l[i,j] <= x[i,j] <= u[i,j];
```